Microsoft Small Basic

Uvod u programiranje



1. poglavlje

Uvod

Small Basic i programiranje

Racunalno programiranje definira se kao postupak stvaranja racunalnog softvera pomodu programskih
jezika. Bas kao $to mi govorimo i razumijemo engleski, Spanjolski ili francuski jezik, ra€unala razumiju
programe napisane na odredenim jezicima. Ti jezici zovu se programski jezici. U pocetku je postojalo
samo nekoliko programskih jezika i bilo ih je lako nauciti i razumjeti. No s vremenom su racunala

i softver postajali sve sloZeniji, programski jezici brzo su se razvijali i obuhvadéali sve kompleksnije
koncepte. Moderni programski jezici zato su prilicno zahtjevni, osobito za pocetnike. To je mnoge
obeshrabrilo i odvratilo od uéenja i rada u podrucju ra¢unalnog programiranja.

Small Basic je programski jezik osmisljen za iznimno jednostavno programiranje, pristupacan je i zabavan
Cak i za pocetnike. Small Basic rusi barijere i sluZi kao uvod u cudesni svijet racunalnog programiranja.



OKkruZenje programa Small Basic
Krenimo s kratkim uvodom u okruZenje Small Basic. Kada pokrenete program Small Basic, otvorit ¢e se
prozor prikazan na sljedecoj slici.

Slika 1 — OkruZenje programa Small Basic

To je okruZenje programa Small Basic u kojem ¢emo pisati i izvrSavati programe napisane na jeziku
Small Basic. OkruZenje se sastoji od nekoliko elemenata oznacenih brojevima.

U uredivacu, oznacenom brojem [1], piSemo programe na jeziku Small Basic. Kada otvorite primjer
programa ili prethodno spremljeni program, prikazat ¢e se u uredivacu. Zatim ga moZete izmijeniti
i spremiti za kasniju upotrebu.

Istodobno mozZete otvoriti i raditi na viSe programa. Svaki program na kojem radite prikazat ce se
u zasebnom uredivacu. Uredivac koji sadrZi program na kojem trenutacno radite naziva se aktivni
uredivac.

Alatna traka, oznacena brojem [2], upotrebljava se za izdavanje naredbi u aktivnom uredivacu
ili okruZzenju. S naredbama na alatnoj traci upoznat cete se postupno u ovom prirucniku.

Radna povrsina, oznacena brojem [3], mjesto je na kojem se prikazuju svi prozori uredivaca.



Nas prvi program
Sada ste se upoznali s okruzenjem programa Small Basic pa moZemo zapoceti s programiranjem
u njemu. Kao $to smo vec rekli, programe piSemo u uredivacu. Dakle, upiSimo sljededi redak u uredivac.

TextWindow.WriteLine("Pozdrav svima")

To je nas prvi program na jeziku Small Basic. Ako sve to¢no upiSete, trebali biste vidjeti nesto sli¢no kao

na slici u nastavku.

Slika 2 — Prvi program

Napisali smo prvi program; pokrenimo ga i pogledajmo Sto ¢e se dogoditi. Program moZemo pokrenuti
klikom na gumb Pokreni na alatnoj traci ili pritiskom na tipku precaca, F5, na tipkovnici. Ako sve bude
iSlo po planu, rezultat pokretanja programa izgledat ée kako je prikazano u nastavku.

Poz= suima
Press any key to continue...

Slika 3 — Rezultat prvog programa

Slika 4 - Intellisense

Spremanje programa

Ako Zelite zatvoriti Small Basic i kasnije raditi na programu koji ste upravo napisali, moZete ga
spremiti. Preporucuje se da povremeno spremite programe kako ne biste izgubili podatke u slucaju
nepredvidenog zatvaranja programa ili nestanka struje. Trenutacni program moZete spremiti klikom
na ikonu "Spremi" na alatnoj traci ili tipkovnim precacem "Ctrl+S" (pritisnite tipku S dok drzite
pritisnutom tipku Ctrl).



2. poglavlje
Analiza naSeg prvog programa

Sto je zapravo ra¢unalni program?

Program je niz uputa za racunalo. Te upute racunalu kazu Sto tocno treba uciniti i ono ih vjerno slijedi.
Jednako kao i ljudi, ra¢unala mogu slijediti upute samo ako su napisane na jeziku koji razumiju. Ti jezici
zovu se programski jezici. Ra¢unalo razumije brojne jezike i Small Basic je jedan od njih.

Zamislite da razgovarate s prijateljem. Vi i vas prijatelj upotrebljavate rijeCi organizirane u recenice
kako biste jedno drugome prenijeli informacije. Jednako tako, programski jezici sadrze zbirke rijeci koje
se mogu organizirati u reenice koje racunalu prenose informacije. Programi su, dakle, skupovi recenica
(ponekad samo nekoliko njih, ponekad vise tisuéa) koje razumiju i programer i racunalo.

Programi na jeziku Small Basic Racunalo razumije brojne jezike. Java,

Program na jeziku Small Basic obi¢no se sastoji od
grupe naredaba. Svaki je redak programa jedna
naredba, a svaka je naredba uputa za racunalo.

C++, Python, VB itd. napredni su moderni

programski jezici koji se upotrebljavaju za

stvaranje jednostavnih i sloZenih programa.

Kada od racunala zatrazimo da izvrsi program na

jeziku Small Basic, ono otvara program i ¢ita prvu naredbu. Razumije Sto od njega trazimo i izvrSava nasu
naredbu. Kada izvrsi nasu prvu naredbu, vraca se na program te Cita i izvrSava drugi redak. Nastavlja

s tim postupkom dok ne dode do posljednjeg retka. Tu je kraj programa.



Vratimo se nasem prvom programu
Napisali smo sljedeci program:

TextWindow.WriteLine("Pozdrav svima")

To je vrlo jednostavan program koji se sastoji od samo jedne naredbe. Ona nalaze racunalu da
u tekstnom prozoru ispise redak teksta Pozdrav svima.

Ta naredba na racunalnom jeziku znaci doslovno:
Write Pozdrav svima

Mozda ste vec¢ primijetili da se naredba sastoji od nekoliko manjih dijelova bas$ kao $to se recenice
sastoje od rijeci. Prva naredba sastoji se od 3 dijela:

a) TextWindow
b) Writeline
c) "Pozdrav svima"

Tocka, zagrade i navodnici znakovi su koji se moraju postaviti na odgovaraju¢a mjesta kako bi racunalo
moglo razumjeti nasu naredbu.

MoZda se sjecate crnog prozora koji se pojavio pri pokretanju naseg prvog programa. Taj crni prozor
zove se TextWindow, ponekad se upotrebljava i naziv "konzola". U njemu se vide rezultati programa.
U naSem programu TextWindow se zove objekt. Programi nam stavljaju na raspolaganje viSe takvih
objekata. Na njima moZemo izvrsiti nekoliko razlicitih operacija. U svom programu veé smo koristili
operaciju WriteLine. MoZda ste primijetili da nakon operacije WriteLine slijedi tekst Pozdrav svima

u navodnicima. Taj tekst prosljeduje se operaciji

WriteLine, koja ga ispisuje korisniku. To se zove Znakovi poput navodnika, tocaka i zagrada
unos za operaciju. Neke operacije zahtijevaju vrlo su vazni u raunalnom programu. Njihov

jedan ili vise unosa, neke nijedan. poloZaj i broj moZe promijeniti znacenje izraza.




Nas drugi program
Upoznali smo osnove svog prvog programa i sada ga moZzemo uljepsati dodavanjem boja.

TextWindow.ForegroundColor = "Yellow"
TextWindow.WriteLine("Pozdrav svima")

Slika 5 — Dodavanje boja

Kada pokrenete program, izraz "Pozdrav svima" ispisat ¢e u prozoru TextWindow, ali ovaj put Zutim

slovima umjesto sivih.

B C\Users\mjperez\AppData\Local\Temp'tmp

Pozdrav svima
Pressz any key to continue...

Slika 6 — Pozdrav svima u Zutoj boji



Pogledajte novu naredbu koju smo dodali izvornom programu. Sadrzi novu rije¢, ForegroundColor,

uz koju nakon znaka jednakosti stoji vrijednost "Yellow". To znaci da smo objektu ForegroundColor
dodijelili vrijednost "Yellow". Razlika izmedu svojstva ForegroundColor i operacije WriteLine jest

u tome da ForegroundColor ne zahtijeva unose ni zagrade. Umjesto toga nakon svojstva slijedi znak
jednako irijec. ForegroundColor definiramo kao svojstvo objekta TextWindow. Slijedi popis vrijednosti
koje su valjane za svojstvo ForegroundColor. Pokusajte zamijeniti "Yellow" s jednom od sljedecih
vrijednosti i pogledajte rezultat — ne zaboravite navodnike, oni su obavezni.

3. poglavlje
Uvod u varijable

Koristenje varijabli u naSem programu

Zar ne bi bilo zgodno kad bi nas program mogao reci "Pozdrav" s imenom korisnika umjesto opcenite
fraze "Pozdrav svima"? Da bi se to postiglo, najprije od korisnika moramo zatraziti njegovo ime, spremiti
ga, a zatim ispisati "Pozdrav" s imenom korisnika. Pogledajmo kako to postici:

TextWindow.Write("Unesite svoje ime: ")
ime = TextWindow.Read()
TextWindow.WriteLine("Pozdrav,"'

+ ime)

Kad upisete i izvrsite taj program, prikazat ¢e se rezultat slican ovom:

0 ChlUsers\mjperez\AppData\Local\TempitrmpD2F tmp.exe _

lInesite svoje ime:

Slika 7 - Zatrazite od korisnika da unese ime

A kad unesete svoje ime i pritisnete ENTER, prikazat ce se sljedece:



0 ChlUsers\mjperez\AppData\Local\TempitrmpD2F tmp.exe _

linesite svoje ime: Uijave
Pozdrav . Uijaye
Press any key to continue...

Slika 8 - Topli pozdrav

Ako ponovno pokrenete program, ponovno ¢e vam se postaviti isto pitanje. MoZete upisati drugo ime,
a racunalo ¢ée vas pozdraviti s tim imenom.

Analiza programa
U programu koji ste upravo pokrenuli, mozda vam je paznju privukao sljededi redak:

ime = TextWindow.Read()

Read() izgleda bas kao WriteLine(), ali bez unosa. To je operacija koja upucuje racunalo da ¢eka dok
korisnik nesto ne upise i pritisne tipku ENTER. Kad korisnik pritisne tipku ENTER, operacija uzima unos
koji je korisnik upisao i vraca ga u program. Ono

Sto je korisnik upisao sada je spremljeno u

varijabli pod nazivom ime. Varijabla se definira Write, bas kao i WriteLine, jos je jedna

kao mjesto gdje moZete privremeno spremiti operacija u prozoru ConsoleWindow.

vrijednosti i kasnije ih koristiti. U gornjem retku Operacija Write omogucuje vam da upisete

nesto u prozor ConsoleWindow, ali omogucuje

varijabla ime koristi se za spremanje imena

korisnika. tekstu u nastavku da se prikaZe u istom retku

kao trenutni tekst.

Zanimljiv je i sljededi redak:

TextWindow.WriteLine("Pozdrav," + ime)

To je mjesto gdje koristimo vrijednost spremljenu u varijabli ime. Uzimamo vrijednost varijable ime,
dodajemo je izrazu "Pozdrav," i piSemo je u prozoru TextWindow.

Nakon postavljanja varijablu moZete koristiti koliko god puta Zelite. Na primjer, moZete napraviti
sljedece:

TextWindow.Write("Unesite svoje ime: ")
ime = TextWindow.Read()

TextWindow.Write("Pozdrav,'
TextWindow.WriteLine("Kako ste,

+ ime + ". ")
"+ ime + "?"



Prikazat ée se sljededi izlazni rezultat:

# CAUsers\mjperez\AppData\Local\Temp\tmp6F8D.tmp.exe _ E

Unezsite svoje ime: Uijaye
Pozdrav.Uijave. Kako ste. Wijaye?
Press any key to continue...

Slika 9 - Ponovno koristenje varijable

Pravila za davanje naziva varijablama
Varijablama se pridruzuju nazivi po kojima ih prepoznajete. Postoje odredena jednostavna pravila i neke
uistinu dobre smjernice za davanje naziva tim varijablama. To su sljedeca pravila:

1. Naziv mora zapocinjati slovom i ne smije se preklapati s nekom od klju¢nih rijeci kao sto su if,
for, then itd.

2. Naziv moze sadrzavati bilo koju kombinaciju slova, znamenki i podvlaka.

3. Korisno je da nazivi varijabli imaju odgovarajuée znacenje. Bududi da duljina naziva varijabli nije
ogranicena, koristite ih da biste opisali njihovu svrhu.

Igranje s brojevima
Upravo smo vidjeli kako pomodu varijabli spremiti ime korisnika. U nekoliko sljedecih programa vidjet
¢emo kako spremiti brojeve u varijablama i kako rukovati njima. Zapo€nimo s vrlo jednostavnim

programom:
brojl = 10
broj2 = 20
broj3 = brojl + broj2

TextWindow.WriteLine(broj3)

Kad pokrenete taj program, dobit cete sljedeci izlazni rezultat:

B C\Users\mjperez\AppData\Local\Temp\tmp2D31.tmp.exe _ C=rea X

30
Press any key to continue...

Slika 10 — Zbrajanje dva broja



U prvom retku programa varijabli broj1 Drimietit cete da broievi nisy okruseni
dodjeljujete vrijednost 10. U drugom retku rlm/:jetft'ceteNa :;)j?\/'l n‘lsu ° ruz:nf
varijabli broj2 dodjeljujete vrijednost 20. navodnicima. Navodnici nisu potrebni za

brojeve. Oni su potrebni samo ako koristite tekst.

U tre¢em retku zbrajate brojl i broj2, a
zatim zbroj dodjeljujete varijabli broj3.
Dakle, u tom ée slucaju broj3 imati vrijednost 30. A evo Sto se ispisuje u prozoru TextWindow.

Izmijenimo malo sada taj program i pogledajmo rezultate:

brojl = 10
broj2 = 20
broj3 = brojl * broj2

TextWindow.WriteLine(broj3)

Gornji program mnotzi varijablu brojl s varijablom broj2, a zatim sprema rezultat u varijabli broj3.
Rezultat tog programa mozete vidjeti u nastavku:

B C\Users\mjperez\AppData\Local\Temp\tmpD4E8.tmp.exe _

208
Press any key to continue...

Slika 11 — MnoZenje dva broja

Isto tako moZete oduzimati ili dijeliti brojeve. Evo primjera za oduzimanje:

broj3 = brojl - broj2

A za dijeljenje se koristi znak "/". Program izgleda ovako:

broj3 = brojl / broj2

A ovo je rezultat dijeljenja:

# ' C\Users\mjperez\AppData\Local\Temp\tmp7ECL.tmp.exe _

A .5
Presszs any key to continue...

Slika 12 - Dijeljenje dva broja



Jednostavan pretvornik temperature

5(°F-32)

U sljedecem ¢emo programu pomocéu formule °C = pretvoriti temperature u stupnjevima

Fahrenheita u temperature u stupnjevima Celzija.

Najprije od korisnika moramo dobiti temperaturu u stupnjevima Fahrenheita te je spremiti
u varijabli. Postoji posebna operacija koja nam omogucuje da proc¢itamo brojeve korisnika,
a to je TextWindow.ReadNumber.

TextWindow.Write("Unesite temperaturu u fahrenheitima: ")
fahr = TextWindow.ReadNumber ()

Nakon Sto smo temperaturu u stupnjevima Fahrenheita spremili u varijabli, moZemo je pretvoriti
u stupnjeve Celzija na sljedeci nacin:

celzij = 5 * (fahr - 32) / 9

Zagrade upucuju racunalo da najprije izracuna dio fahr — 32, a zatim ostatak. Nakon toga samo moramo

prikazati rezultat korisniku. Kad sve to spojimo, dobivamo ovaj program:

TextWindow.Write("Unesite temperaturu u fahrenheitima: ")

fahr = TextWindow.ReadNumber()

celzij = 5 * (fahr - 32) / 9

TextWindow.WriteLine("Temperatura u stupnjevima Celzija iznosi
+ celzij)

n

A ovo je rezultat tog programa:

0 C\Users\mjperez\AppData\Local\Temp\tmpAEE6.tmp.exe

lInesite temperaturu u fahrenheitima: 77
Temperatura u stupnjevima Celzija iznosi 25
Press any key to continue...

Slika 13 - Pretvorba temperature



4. poglavlje
Uvjeti i grananje

Ako se vratimo na nas prvi program, zar ne bi bilo sjajno da, umjesto opcenitog Pozdrav svima, mozemo
reci Dobro jutro svima ili Dobar dan svima ovisno o dobu dana? U nasem sljede¢em programu postici
¢emo da racunalo kaze Dobro jutro svima prije 12 sati, a Dobar dan poslije 12 sati.

If (Clock.Hour < 12) Then
TextWindow.WritelLine("Dobro jutro svima")

EndIf

If (Clock.Hour >= 12) Then
TextWindow.WritelLine("Dobar dan svima")

EndIf

Ovisno o tome kada se program pokrene, prikazuje se od sljedecih izlaznih rezultata:

# CAUsers\mjperez\AppDatailecal\TempitmpCD1l.tmp.exe _

Dobro jutro svima
Press any key to continue...

Slika 14 - Dobro jutro svima



B ChlUsers\mjperez’\AppDatat\Local\Templtmp2254.tmp.exe _

Dobar dan svima
Presz any key to continue...

Slika 15 - Dobar dan svima

Analizirajmo prva tri retka programa. Ve¢ ste shvatili da taj redak upucuje racunalo da se, ako Clock.Hour
prikazuje da je ranije od 12 h, ispiSe "Dobro jutro svima". Izrazi If, Then i EndIf posebni su izrazi koje
ra¢unalo razumije nakon pokretanja programa. Nakon If uvijek slijedi uvjet, a to je u ovom slucaju
(Clock.Hour < 12). Ne zaboravite da su zagrade

potrebne da bi raunalo shvatilo vase namjere. U programskom jeziku Small Basic za

Nakon uvjeta slijedi then i stvarna operaciju koju pristupanje trenutnom datumu i vremenu

je potrebno izvrsiti. A nakon operacije dolazi EndIf. koristi se objekt Clock. On vam nudi i velik
To racunalo upucuje da je uvjetno izvrSenje broj svojstava pomocu kojih moZete zasebno

gotovo. otvoriti trenutni dan, mjesec, godinu, sat,

minute i sekunde.

Izmedu then i EndIf mozZe postojati viSe operacija,
a racunalo ¢e ih sve izvrsiti ako je uvjet valjan.
Na primjer, moZete napisati neSto poput ovoga:

If (Clock.Hour < 12) Then
TextWindow.Write("Dobro jutro. ")
TextWindow.WritelLine("Jeste 1i doruckovali?")

EndIf

Izraz Else

U programu na pocetku ovog poglavlja mozda ste primijetili da je drugi uvjet nekako suvisan. Vrijednost
Clock.Hour moze biti manja od 12 ili ne. Zapravo nije ni bilo potrebno izvoditi drugu provjeru. U takvim
sluajevima moZemo skratiti dvije naredbe if..then..endif u jednu pomocu novog izraza else.

Kad bismo ponovno napisali taj program pomocu else, evo kako bi izgledao:

If (Clock.Hour < 12) Then
TextWindow.WriteLine("Dobro jutro svima")
Else
TextWindow.WriteLine("Dobar dan svima")
EndIf

Taj ¢e program imati potpuno istu funkciju kao onaj prvi pa dolazimo do vrlo vazne lekcije u vezi
s racunalnim programiranjem:



u programiranju obicno postoji puno nacina da se napravi ista stvar. Ponekad jedan nacin ima
vise smisla od drugog. Odabir je na programeru. Kako budete pisali sve vise programa i stjecali
iskustvo, pocet cete primjecivati te razlicite tehnike i njihove prednosti i nedostatke.

Uvlake

U svim primjerima mozete vidjeti da su naredbe izmedu If, Else i EndIf uvucene. Uvlake nisu potrebne.
Racunalo ¢e bez problema razumjeti program i bez njih. No one nam pomazu da lakse vidimo i
razumijemo strukturu programa. Stoga se obi¢no smatra dobrom praksom uvuci naredbe izmedu
takvih blokova.

Parno ili neparno
Sad kad smo svladali naredbu If..Then..Else..EndlIf, napiSimo program koji ¢e odrediti je li neki broj paran
ili neparan.

TextWindow.Write("Unesite broj: ")
broj = TextWindow.ReadNumber ()
ostatak = Math.Remainder(broj, 2)
If (ostatak = @) Then
TextWindow.WritelLine("Broj je paran")
Else
TextWindow.WriteLine("Broj je neparan")
EndIf

Kad pokrenete taj program, prikazat ce se izlazni rezultat sli¢an ovom:

0 C\Users\mjperez\AppData\Local\Temp\tmp3AB4.tmp.exe

Unesite broj: 32485
Broj je neparan
Press any key to continue...

Slika 16 - Parno ili neparno

U tom smo programu uveli jos jednu korisnu operaciju, Math.Remainder. | da, kao sto ste mozda veé
shvatili, Math.Remainder dijeli prvi broj s drugim, a zatim vraca ostatak.

Grananje

Ne zaboravite, u drugom poglavlju naucili ste da racunalo obraduje jednu po jednu naredbu programa
i to redoslijedom od gore prema dolje. No postoji posebna naredba koja racunalo upucuje da preskoci
na drugu naredbu bez obzira na redoslijed. Pogledajmo sljedeci program.



i=1
start:
TextWindow.WritelLine(i)
i=1i+1
If (i < 25) Then

Goto start
EndIf

Slika 17 - Koristenje naredbe Goto

U gornjem programu varijabli i dodijelili smo vrijednost 1. Potom smo dodali novu naredbu koja zavrsava
dvotockom (:)

start:

To se zove oznaka. Oznake su kao knjizne oznake koje racunalo razumije. KnjiZznoj oznaci mozZete dati
kakvo god ime Zelite i u program mozete dodati koliko god oznaka Zelite, pod uvjetom da sve imaju
jedinstveni naziv.

Evo jos jedne zanimljive naredbe:

Time se samo racunalo upucuje da doda 1 varijabli i te da taj zbroj ponovno dodijeli varijabli i. Dakle,
ako je vrijednost varijable i bila 1 prije te naredbe, nakon njezina izvodenja te ¢e vrijednost biti 2.

| konacno,



If (i < 25) Then
Goto start
EndIf

To je dio koji upuéuje racunalo da, ako je vrijednost varijable i manja od 25, pocne izvrSavati naredbe iz
knjizne oznake start.

Neograniceno izvrsavanje

Pomocu naredbe Goto moZete postiéi da racunalo ponavlja nesto neogranicen broj puta. Na primjer,
mozZete uzeti program Parno ili neparno te ga izmijeniti na nacin prikazan u nastavku i program ce se
neprestano izvoditi. Program moZete zaustaviti klikom na gumb Zatvori (X) na alatnoj traci u gornjem
desnom kutu prozora.

pocetak:
TextWindow.Write("Unesite broj: ")
broj = TextWindow.ReadNumber ()
ostatak = Math.Remainder(broj, 2)
If (ostatak = @) Then
TextWindow.WriteLine("Broj je paran")
Else
TextWindow.WriteLine("Broj je neparan")
EndIf
Goto pocetak

0 C\Users\mjperez\AppData\Local\Temp'\tmpSEAT.tmp.exe _

Slika 18 — Neograniceno izvodenje programa Parno ili neparno



5. poglavlje

Petlje

Petlja For
Pogledajmo program koji smo napisali u prethodnom poglavlju.

i=1
start:
TextWindow.WritelLine(i)
i=1i+1
If (i < 25) Then

Goto start
EndIf

Taj program ispisuje brojeve od 1 do 24 po redu. Taj proces postupnog povecéavanja varijable vrlo je
uobicajen u programiranju pa programski jezici obi¢no omogucuju jednostavniji nacin da se to napravi.
Gornji program ekvivalentan je programu u nastavku.

For i =1 To 24
TextWindow.WritelLine(i)
EndFor



A izlazni je rezultat sljededi:

W | C\Users\mjperez\AppData\Loca I\Temp\tmpTSFD.tmp.eme_

Presz any key to continue...

Slika 19 — Koristenje petlje For

Primijetit ¢ete da smo program od 8 redaka smanijili na program od 4 retka, a da ipak radi potpuno istu
stvar kao program od 8 redaka! Sjeéate se da smo ranije rekli da obi¢no postoji nekoliko nacina da se
napravi ista stvar? To je sjajan primjer za to.

U programiranju se For..EndFor naziva petl/ja. Ona vam omoguduje da uzmete varijablu, date joj pocetnu
i zavrSnu vrijednost i pustite racunalo da umjesto vas postupno povecava tu varijablu. Svaki put kad
racunalo postupno poveca varijablu, izvodi naredbe izmedu petlji For i EndFor.

Ali ako Zelite da se varijabla poveca za 2, a ne za 1 jer, na primjer, Zelite ispisati sve neparne brojeve
izmedu 1i 24, to moZete napraviti pomocu petlje.

For i = 1 To 24 Step 2
TextWindow.WriteLine(i)
EndFor

CA\Users\mjperez\AppData‘\Local\Temp\tmpBACE.tmp.exe _LI_IéJ

Press any key to continue...

Slika 20 - Samo neparni brojevi



Dio Step 2 naredbe For upucuje racunalo da poveca vrijednost varijable i za 2 umjesto uobicajenog 1.
Pomocu Step moZete odrediti svako postupno povecanje koje Zelite. MoZete ¢ak odrediti i negativnu
vrijednost za korak te postiéi da racunalo broji unatrag, kao u primjeru u nastavku:

For i = 10 To 1 Step -1
TextWindow.WriteLine(i)
EndFor

n C:\Users\mjperez\AppData\LocaI\Temp\tmngM.tmp.exe_ —n Eh

ress any key to continue...

Slika 21 - Brojenje unatrag

Petlja While

Petlja While jos$ je jedna metoda za izradu petlje, posebno ako broj petlje nije unaprijed poznat.
Dok se petlja For izvodi unaprijed odredeni broj puta, petlja While izvodi se dok se ne ispuni zadani
uvjet. U primjeru u nastavku broj dijelimo napola sve dok je rezultat veéi od jedan.

broj = 100

While (broj > 1)
TextWindow.WriteLine(broj)
broj = broj / 2

EndiWhile

# | C\Usersim) p-erez\AppData\LocaT\Temp\tmp?D?3.h'np._ E

Press any key to continue...

Slika 22 - Petlja za dijeljenje broja napola



U gornjem programu varijabli broj dodjeljujemo vrijednost 100 i izvodimo petlju while sve dok je broj veé
od 1. Unutar petlje ispisujemo broj, a zatim ga dijelimo s dva, ¢ime ga zapravo dijelimo napola. Kao $to
se i o€ekivalo, izlazni rezultat programa brojevi su koji se progresivno jedan za drugim dijele napola.

Bilo bi teSko napisati taj program pomocu petlje For jer ne znamo koliko ¢e se puta petlja izvoditi.
Pomodu petlje While jednostavno je provjeriti uvjet te zatraziti od racunala da nastavi petlju ili da
je zaustavi.

Zanimljivo je primijetiti da se svaka petlja While moZe razviti u naredbu If..Then. Na primjer,
gornji program moZe se ponovno napisati na sljedeci nacin, a da to ne utjee na krajnji rezultat.

broj = 100

startLabel:
TextWindow.WriteLine(broj)
broj = broj / 2

If (broj > 1) Then
Goto startLabel
EndIf

Zapravo, racunalo interno ponovno zapisuje
svaku petlju While u naredbe koje koriste

If..Then zajedno s jednom ili vise naredbi Goto.




6. poglavlje
Pocetak crtanja

Dosad smo u nasim primjerima pomocdu prozora TextWindow objasnjavali osnove jezika Small Basic.
No Small Basic ima napredan skup mogucnosti crtanja koje éemo poceti istrazivati u ovom poglavlju.

Uvod u GraphicsWindow

Bas kao S$to smo imali prozor TextWindow koji nam je omogucio da radimo s tekstom i brojevima,
Small Basic nudi i prozor GraphicsWindow pomocu kojeg mozZemo crtati. Za pocetak ¢emo prikazati
prozor GraphicsWindow.

GraphicsWindow.Show()

Prilikom pokretanja tog programa primijetit ¢éete da se umjesto uobicajenog crnog prozora s tekstom
prikazuje bijeli prozor, kao Sto je ovaj prikazan u nastavku. Zasad u tom prozoru ne moZzemo nista
posebno raditi. Ali to ¢e biti osnovni prozor u kojem ¢emo raditi u ovom poglavlju. MoZete ga zatvoriti
klikom na gumb "X" u gornjem desnom kutu prozora.



# ' SmallBasic Graphics WindowA

Slika 23 - Prazan graficki prozor

Postavljanje grafickog prozora
Graficki prozor omoguduje vam da prilagodite njegov izgled prema svojim Zeljama. Mozete promijeniti
naslov, pozadinu i njegovu veli¢inu. Izmijenimo ga sada malo, tek toliko da bismo se upoznali s njim.

GraphicsWindow.BackgroundColor = "SteelBlue"
GraphicsWindow.Title "Moj graficki prozor"
GraphicsWindow.Width 320
GraphicsWindow.Height = 200
GraphicsWindow.Show()



Evo kako izgleda prilagodeni graficki prozor. Boju pozadine moZete promijeniti u bilo koju od puno
vrijednosti navedenih u Dodatku B. Poigrajte se s tim svojstvima da biste vidjeli kako se moZe izmijeniti
izgled prozora.

Slika 24 - Prilagodeni graficki prozor

Crtanje linija

Nakon $to otvorimo prozorGraphicsWindow, u njemu moZemo crtati oblike, tekst, pa ¢ak i slike.
Zapocnimo s crtanjem nekih jednostavnih oblika. Evo programa koji to crta nekoliko linija u grafickom
prozoru.

GraphicsWindow.Width = 200
GraphicsWindow.Height = 200
GraphicsWindow.DrawLine (10, 10, 100, 100)
GraphicsWindow.DrawLine (10, 100, 100, 10)

B SmallBasic G.. s =l e

Slika 25 — Prekrizene linije



Prva dva retka programa sluze za postavljanje
Umjesto naziva za boje moZete koristiti oznake

web-boja (#RRGGBB). Na primjer, #FF0000
oznacava crvenu, #FFFFOO Zutu itd. Vise o

prozora, a druga dva retka crtaju prekrizene

linije. Prva dva broja koja slijede nakon DrawLine
odreduju pocetne x i y koordinate, a druga dva
odreduju zavrsne x i y koordinate. Zanimljiva stvar

bojama saznat ¢emo u [TODO poglavlju Boje]

kod racunalnog crtanja to je Sto koordinate (0, 0)
zapocinju u gornjem lijevom kutu prozora. Zbog toga se u koordinatnom prostoru uzima da se prozor
nalazi u drugom kvadrantu.

81 SmallBasic G... s o= e S

Slika 26 - Koordinatna mapa

Ako se vratimo na program za crtanje linija, zanimljivo je primijetiti da vam Small Basic omoguduje da
izmijenite svojstva linije, kao Sto su njezina boja i debljina. Najprije ¢emo izmijeniti boju linija kao sto
je prikazano u programu u nastavku.

GraphicsWindow.Width = 200
GraphicsWindow.Height = 200

GraphicsWindow.PenColor = "Green"
GraphicsWindow.DrawlLine (10, 10, 100, 100)
GraphicsWindow.PenColor = "Gold"

GraphicsWindow.DrawLine (10, 100, 100, 10)

i SmanBasicG...[:' en X

Slika 27 — Promjena boje linije



Sada ¢éemo izmijeniti i njezinu veli¢inu. U programu u nastavku mijenjamo Sirinu linije u 10, umjesto
zadane linije koja je 1.

GraphicsWindow.Width = 200
GraphicsWindow.Height = 200

GraphicsWindow.PenWidth = 10
GraphicsWindow.PenColor = "Green"
GraphicsWindow.DrawlLine(10, 10, 100, 100)
GraphicsWindow.PenColor = "Gold"

GraphicsWindow.DrawLine (10, 100, 100, 10)

»  SmallBasic G.. = (| e

Slika 28 - Debele linije u bojama

Pomocu operacija PenWidth i PenColor mijenja se olovka kojom se iscrtavaju linije. One ne utjeCu samo
na linije, vec i na svaki oblik koji se crta nakon azuriranja svojstava.

Pomocu naredbi petlji o kojima smo govorili u prethodnim poglavljima moZemo jednostavno napisati
program koji iscrtava vise linija sa sve debljom olovkom.

GraphicsWindow.BackgroundColor = "Black"
GraphicsWindow.Width = 200
GraphicsWindow.Height = 160
GraphicsWindow.PenColor = "Blue"

For i =1 To 10
GraphicsWindow.PenWidth = i
GraphicsWindow.DrawlLine(20, i * 15, 180, i * 15)
endfor

Slika 29 - Razlicite Sirine olovke

Zanimljivi dio tog programa jest petlja u kojoj povec¢avamo PenWidth prilikom svakog pokretanja petlje,
a zatim crtamo novu liniju ispod stare.

Crtanje i popunjavanje oblika

Kad je rijec o crtanju oblika, obi¢no postoje dvije vrste operacija za svaki oblik. To su operacije

Draw (crtanje) i Fill (popunjavanje). U operacijama crtanja pomocu olovke se iscrtava obris oblika,

a u operacijama popunjavanja oblik se boji pomocu kista. U programu u nastavku, primjerice, imamo
dva pravokutnika. Jedan je nacrtan pomocu crvene olovke, a drugi popunjen pomocu zelenog kista.

GraphicsWindow.Width = 400
GraphicsWindow.Height = 300

GraphicsWindow.PenColor= "Red"
GraphicsWindow.DrawRectangle (20, 20, 300, 60)

GraphicsWindow.BrushColor = "Green"
GraphicsWindow.FillRectangle(60, 100, 300, 60)

Slika 30 — Crtanje i popunjavanje

Da bi se nacrtao ili popunio pravokutnik, potrebna su Cetiri broja. Prva dva predstavljaju X-koordinatu

i Y-koordinatu za gorniji lijevi kut pravokutnika. Treci broj odreduje Sirinu pravokutnika, a ¢etvrti njegovu
visinu. Zapravo, isto se primjenjuje na crtanje i popunjavanje elipsa, kao Sto je prikazano u programu

u nastavku.

GraphicsWindow.Width = 400
GraphicsWindow.Height = 300

GraphicsWindow.PenColor= "Red"
GraphicsWindow.DrawEllipse(20, 20, 300, 60)

GraphicsWindow.BrushColor = "Green"
GraphicsWindow.FillEllipse(60, 100, 300, 60)



| & SmallBasic Graphics Win

Slika 31 - Crtanje i popunjavanje elipsa



Elipse su samo varijanta krugova. Ako Zelite crtati krugove, morate odrediti istu Sirinu i visinu.

GraphicsWindow.Width = 400
GraphicsWindow.Height = 300

GraphicsWindow.PenColor= "Red"
GraphicsWindow.DrawEllipse(20, 20, 100, 100)

GraphicsWindow.BrushColor = "Green"
GraphicsWindow.FillEllipse(100, 100, 100, 100)

i SmallBasic Graphics Windo

Slika 32 - Krugovi



7. poglavlje
Zabava uz oblike

U ovom éemo se poglavlju zabaviti uz sve $to smo naucili dosad. U njemu se nalaze primjeri koji
prikazuju neke zanimljive nac¢ine kombiniranja svega naucenog dosad da bi se izradili neki zgodni
programi.

Mnostvo pravokutnika
Ovdje crtamo vise pravokutnika u petlji i svaki je veci od prethodnog.

GraphicsWindow.BackgroundColor = "Black"
GraphicsWindow.PenColor = "LightBlue"
GraphicsWindow.Width = 200
GraphicsWindow.Height = 200

For i = 1 To 100 Step 5
GraphicsWindow.DrawRectangle(100 - i, 100 - i, i * 2, i * 2)
EndFor

Slika 33 - Mnostvo pravokutnika

Mnostvo krugova
Varijanta prethodnog programa u kojoj se umjesto kvadrata crtaju krugovi.

GraphicsWindow.BackgroundColor = "Black"
GraphicsWindow.PenColor = "LightGreen"
GraphicsWindow.Width = 200
GraphicsWindow.Height = 200

For i = 1 To 100 Step 5
GraphicsWindow.DrawEllipse(100 - i, 100 - i, i * 2, i * 2)
EndFor

»  SmallBasic G... = bl e

Slika 34 — Mnostvo krugova



Koristenje slucajne boje

Ovaj program pomocu operacije GraphicsWindow.GetRandomColor odreduje slucajne boje za kist,

a zatim pomocu operacije Math.GetRandomNumber odreduje x- i y-koordinate za krugove. Te se dvije
operacije mogu kombinirati na zanimljive nacine da bi se izradili interesantni programi koji daju razli¢ite

rezultate prilikom svakog izvodenja.

GraphicsWindow.BackgroundColor = "Black"
For i = 1 To 1000
GraphicsWindow.BrushColor = GraphicsWindow.GetRandomColor ()
X = Math.GetRandomNumber (640)
y = Math.GetRandomNumber (480)
GraphicsWindow.FillEllipse(x, y, 10, 10)
EndFor

Slika 35 — Koristenje slu¢ajne boje



Fraktali

U sljede¢em programu moguce je nacrtati jednostavan fraktal u obliku trokuta pomocu slucajnih
brojeva. Fraktal je geometrijski oblik koji je moguce podijeliti na dijelove, pri ¢emu svaki od dijelova

u potpunosti nalikuje nadredenom obliku. U ovom sluc¢aju program crta stotine trokuta od kojih svaki
nalikuje svom nadredenom trokutu. A bududéi da se program izvodi nekoliko sekundi, mozZete uistinu
vidjeti kako trokuti nastaju iz obic¢nih toc¢aka. Samu logiku ponesto je teSko opisati pa ¢emo vam
prepustiti da to sami istrazite za vjezbu.

GraphicsWindow.BackgroundColor = "Black"
X = 100
y = 100

For i = 1 To 100000
r = Math.GetRandomNumber(3)
ux = 150
uy = 30
If (r = 1) then
ux = 30
uy = 1000
EndIf

If (r = 2) Then
ux = 1000
uy = 1000
EndIf

X
y

(x + ux) / 2
(y +uy) /2

GraphicsWindow.SetPixel(x, y, "LightGreen")
EndFor



-

EF_

e ]

digiay

i

4

[

By

Slika 36 - Crtanje fraktala u obliku trokuta

Ako uistinu Zelite vidjeti kako tocke polako oblikuju fraktal, mozete u petlju uvesti odgodu pomodu
operacije Program.Delay. Ova operacija uzima broj kojim se u milisekundama navodi kolika ¢e biti

odgoda. Slijedi izmijenjeni program, a izmijenjeni redak prikazan je podebljano.



GraphicsWindow.BackgroundColor = "Black"
X = 100
y = 100

For i = 1 To 100000
r = Math.GetRandomNumber(3)
ux = 150
uy = 30
If (r = 1) then
ux = 30
uy = 1000
EndIf

If (r = 2) Then
ux = 1000
uy = 1000
EndIf

X
y

(x + ux) / 2
(y +uy) /2

GraphicsWindow.SetPixel(x, y, "LightGreen")
Program.Delay(2)
EndFor

Poveéanjem odgode program e se usporiti. Poigrajte se razli¢itim duljinama trajanja da biste vidjeli Sto
najbolje odgovara vasem ukusu.

Jos jedna izmjena koju moZete napraviti u ovom programu zamjena je sljedeceg retka:

GraphicsWindow.SetPixel(x, y, "LightGreen")

ovim retkom

boja = GraphicsWindow.GetRandomColor()
GraphicsWindow.SetPixel(x, y, boja)

Zahvaljujudi toj izmjeni program ce crtati piksele trokuta koristeci sluc¢ajne boje.



8. poglavlje
Crtanje pomocu kornjace

Logo

Sedamdesetih godina dvadesetog stoljec¢a postojao je vrlo jednostavan, ali napredan programski jezik
pod nazivom Logo koji je koristilo nekoliko istraZivaca. To je bilo tako dok netko programskom jeziku
nije dodao nesto Sto se zove "crtanje pomocdu kornjace" te time ucinio dostupnom "kornjacu" koja je
bila vidljiva na zaslonu i odgovarala na naredbe kao $to su Idi naprijed, Skreni desno, Skreni lijevo itd.
Pomocu kornjace ljudi su mogli crtati zanimljive oblike na zaslonu. To je programski jezik odmah ucinilo
pristupacnim i privlacnim ljudima svih dobnih skupina te je bilo uvelike odgovorno za njegovu ludu
popularnost osamdesetih godina dvadesetog stoljeéa.

Small Basic ima objekt Turtle s velikim brojem naredbi koje se mogu pozivati unutar Small Basic
programa. U ovom ¢emo poglavlju pomodu kornjace crtati graficke oblike na zaslonu.

Kornjaca
Za pocetak moramo kornjacu uciniti vidljivom na zaslonu. To se moZe posti¢i pomocu jednostavnog
programa koji se sastoji od jednog retka.

Turtle.Show()



Prilikom pokretanja tog programa primijetit éete bijeli prozor, isti kao onaj koji smo vidjeli u prethodnom
poglavlju, osim Sto ovaj ima kornjacu u sredini. Upravo ¢e ta kornjaca pratiti naSe upute i crtati sve sto
zatrazimo od nje.

# ' SmallBasic Graphics Window E x |

Slika 37 - Kornjaca je vidljiva

Pomicanje i crtanje

Jedna od naredbi koju kornjaca razumije jest Move. Ta operacija uzima broj kao ulaznu vrijednost.
On upuduje kornjacu koliko se daleko mora pomaknuti. Recimo, u primjeru u nastavku zatrazit ¢emo
od kornjace da se pomakne za 100 piksela.

Turtle.Move(100)

Prilikom pokretanja tog programa uistinu

¢ete vidjeti kornjacu kako se polako pomice Prilikom koristenja operacija s kornjacom nije
100 piksela prema gore. Dok se pomice, primijetit potrebno pozivati naredbu Show(). Kornjaca ce
¢ete i kako iza sebe povlaéi crtu. Nakon $to se biti automatski vidljiva pri svakom izvrsavanju
kornja&a prestane pomicati, dobiveni rezultat operacije s kornjacom.

izgledat ¢e otprilike kao na slici u nastavku.




#  SmallBasic Graphics Windo

Slika 38 - Pomicanje za stotinu piksela

Crtanje kvadrata

Kvadrat ima Cetiri stranice, dvije okomite i dvije vodoravne. Da bismo nacrtali kvadrat, potrebno je
postici da kornjaca povuce liniju, okrene se desno i povuce jos jednu liniju, a zatim nastavi dok nisu
gotove sve Cetiri stranice. Evo kako to izgleda prevedeno u program.

Turtle.Move(100)
Turtle.TurnRight()
Turtle.Move(100)
Turtle.TurnRight ()
Turtle.Move(100)
Turtle.TurnRight()
Turtle.Move(100)
Turtle.TurnRight()



Kad pokrenete taj program, vidjet ¢ete da kornjaca crta kvadrat, liniju po liniju, a rezultat izgleda kao na
slici u nastavku.

i ' SmallBasic Graphics Wi

Slika 39 - Kornjaca crta kvadrat

Zanimljivo je napomenuti da opetovano dajemo iste dvije naredbe — toc¢no Cetiri puta. Ve¢ smo naucili
da se takve naredbe koje se ponavljaju mogu izvrsiti pomocu petlji. Dakle, ako uzmemo gornji program
i izmijenimo ga tako da koristi petlju For..EndFor, dobit ¢emo puno jednostavniji program.

For i =1 To 4
Turtle.Move(1090)
Turtle.TurnRight()

EndFor



Promjena boja

Kornjaca crta u potpuno istom prozoru GraphicsWindow koji smo vidjeli u prethodnom poglavlju.
To znadi da su sve operacije koje smo upoznali u prethodnom poglavlju i dalje valjane ovdje.

Na primjer, sljedeci program nacrtat ¢e kvadrat sa svakom stranicom u razli¢itoj boji.

For i =1 To 4
GraphicsWindow.PenColor = GraphicsWindow.GetRandomColor ()
Turtle.Move(1090)
Turtle.TurnRight()

EndFor

i ' SmallBasic Graphics Windo

Slika 40 — Promjena boja

Crtanje sloZenijih oblika

Kornjaca, osim operacija TurnRight i TurnLeft izvrSava operaciju Turn. Ta operacija uzima ulaznu
vrijednost kojom se odreduju kut zakretanja. Pomodu te operacije mogucde je nacrtati mnogokut
s bilo kojim brojem stranica. Sljedeci program crta Sesterokut (mnogokut sa Sest stranica).

For i =1 To 6
Turtle.Move(100)
Turtle.Turn(60)

EndFor



Isprobajte taj program da biste vidjeli hoée li uistinu nacrtati Sesterokut. Primijetit ¢ete da koristimo
naredbu Turn(60) jer je kut izmedu stranica 60 stupnjeva. Za takav mnogokut koji ima sve iste stranice,
kut izmedu stranica moze se jednostavno dobiti podjelom broja 360 s brojem stranica. Kad znamo taj
podatak pomodu varijabli mozemo napisati prili¢cno genericki program koji moze nacrtati mnogokut

s bilo kojim brojem stranica.

stranice = 12

duljina = 400 / stranice
kut = 360 / stranice

For i = 1 To stranice
Turtle.Move(duljina)
Turtle.Turn(kut)

EndFor

Pomocu tog programa moZete nacrtati bilo koji mnogokut izmjenom varijable stranice. Ako tu
stavimo 4, dobit ¢emo kvadrat s kojim smo zapoceli. Ako se stavi dovoljno velik broj, na primjer 50,

rezultat bi bilo tesko razlikovati od kruga.

#  SmallBasic Graphics Window

Slika 41 - Crtanje mnogokuta s 12 stranica



Pomodu tehnike koju smo upravo naucili, mozemo postici da kornjaca crta visestruke krugove, ali svaki
put s malim pomakom, Sto daje zanimljiv rezultat.

stranice = 50
duljina = 400 / stranice
kut = 360 / stranice

Turtle.Speed = 9

For j =1 To 20
For i = 1 To stranice
Turtle.Move(duljina)
Turtle.Turn(kut)
EndFor
Turtle.Turn(18)
EndFor

Gornji program ima dvije For..EndFor petlje,

U gornjem programu natjerali smo kornjacu

jednu unutar druge. Unutarnja petlja (i= 1 do

stranica) sliéna je programu za mnogokut i sluZi da se brZe krece postavljanjem brzine na 9.

za crtanje kruga. Vanjska petlja (j = 1 do 20) sluzi To je svojstvo moguce postaviti na bilo koju

za neznatno okretanje kornjage za svaki nacrtani vrijednost izmedu 1i 10 da bi se kretala onoliko

krug. To kornjadi govori da nacrta 20 krugova. Kad brzo koliko Zelite.
se sve to koristi zajedno, ovaj program rezultira

vrlo zanimljivim uzorkom, kao sto je onaj prikazan u nastavku.



"% SmallBasic Graphics Window’ = S

Slika 42 - Vrtnja u krug

Pomicanje uokolo

Ako ne Zelite da kornjaca crta, moZete pozvati operaciju PenUp. To vam omoguduje da pomaknete
kornjacu bilo gdje na zaslonu, a da se pritom ne povlaci linija. Pozivanjem naredbe PenDown kornjaca
¢e ponovno poceti crtati. To se moZze koristiti za postizanje nekih zanimljivih efekata, kao Sto su
istockane linije. Evo programa koji to koristi za crtanje poligona s isto¢kanim linijama.

stranice = 6

duljina = 400 / stranice
kut = 360 / stranice

For i = 1 To stranice
For j =1 To 6
Turtle.Move(duljina / 12)
Turtle.PenUp()
Turtle.Move(duljina / 12)
Turtle.PenDown()
EndFor
Turtle.Turn(kut)
EndFor



Taj program ponovno ima dvije petlje. Unutarnja petlja crta jednu istoCkanu liniju, a vanjska petlja
odreduje koliko ¢e se linija iscrtati. U nasem primjeru koristili smo 6 za varijablu stranice te smo

stoga dobili Sesterokut s istockanim linijama, kao u nastavku.

B8 SmaliBasic Graphics Winde [ )

Slika 43 - KoriStenje naredbi PenUp i PenDown



9. poglavlje
Potprogrami

Cesto se prilikom pisanja programa dogodi da moramo uvijek iznova izvriavati isti skup koraka. U tim
slu¢ajevima vjerojatno ne bi imalo smisla vise puta pisati iste izjave. Tada dobro dodu potprogrami.

Potprogram je dio koda unutar veéeg programa koji obi¢no izvrSava odredenu funkciju i koji se moze
pozvati bilo gdje u programu. Potprogrami se prepoznaju po nazivu koji slijedi iza kljucne rijeci Sub,
a zavrSavaju klju¢nom rijeci EndSub. Na primjer, sljededi isjeCak koda predstavlja potprogram pod
nazivom PrintTime, a izvrSava zadatak ispisa trenutnog vremena za TextWindow.

Sub PrintTime
TextWindow.WriteLine(Clock.Time)
EndSub

U nastavku je prikazan program koji obuhvaca potprogram i poziva ga ss razlicitih mjesta.

PrintTime()

TextWindow.Write("Unesite svoje ime: ")
ime = TextWindow.Read()
TextWindow.Write(ime +
PrintTime()

, sada je ovoliko sati: ")

Sub PrintTime
TextWindow.WriteLine(Clock.Time)
EndSub



I C\Users\mjperez\AppData\Local\Temp'tmpS4

: Uijaye
» sada Je ovoliko sati: 3:58:87 PM
Presz any key to continue...

Slika 44 - Pozivanje jednostavnog potprograma

Potprogram se izvrSava pozivanjem naredbe SubroutineName(). Kao i obi¢no, interpunkcijski znakovi "()"
potrebni su da bi se ra¢unalu reklo da Zelite izvrsiti potprogram.

Prednosti koristenja potprograma
Kao $to smo upravo vidjeli gore, potprogrami omogucuju da se smanji koli¢ina koda koju je potrebno
unijeti. Nakon Sto ste napisali potprogram
PrintTime, moZzete ga pozvati bilo gdje u

Ne zaboravite da potprogram programa

rogramu i on Ce ispisati trenutno vrijeme. . v . .
prog P J SmallBasic moZete pozvati samo unutar istog

programa. Potprogram nije moguce pozvati

Osim toga, potprogrami omogucuju rastavljanje
slozenih problema na jednostavnije dijelove. unutar drugog programa.

Ako, primjerice, morate rijesiti slozenu jednadzbu,

morate napisati nekoliko potprograma koji rjeSavaju manje dijelove slozene jednadzbe. Zatim mozete
spojiti rezultate da biste dobili rjeSenje izvorne sloZene jednadzbe.

Potprogrami mogu i poboljsati Citljivost programa. Drugim rije¢ima, ako imate potprograme

s odgovarajucim nazivima za dijelove programa koji se obi¢no izvode, program ce biti lako Citljiv

i razumljiv. To je vrlo vazno ako Zelite shvatiti program koji je napisao netko drugi ili ako Zelite da
drugi shvate vas program. Ponekad je to korisno ¢ak i ako Zelite procitati vlastiti program, primjerice
tjedan dana nakon Sto ste ga napisali.



Koristenje varijabli

Unutar potprograma mozete pristupiti svim varijablama u programu te ih koristiti. Sljedeéi program,
primjerice, prihvaca dva broja i ispisuje veci od ta dva broja. Primijetit cete da se varijabla maksimalno
koristi unutar potprograma i izvan njega.

TextWindow.Write("Unesite prvi broj: ")
brojl = TextWindow.ReadNumber()
TextWindow.Write("Unesite drugi broj: ")
broj2 = TextWindow.ReadNumber()

FindMax()
TextWindow.WriteLine("Maksimalni je broj:

+ maksimalno)

Sub FindMax
If (brojl > broj2) Then
maksimalno = brojl
Else
maksimalno
EndIf
EndSub

broj2

A izlazni rezultat tog programa izgleda ovako.

o | CAUsers\mjperez\AppData\Local\Temp'tmp
Unesite prvi broj: 334

e drugi broj: 299
imalni je broj: 334
Pressz any key to continue...

Slika 45 - Odredivanje veceg od dva broja pomocu potprograma

Pogledajmo jos jedan primjer koji ilustrira koristenje potprograma. Ovaj ¢emo puta koristiti graficki
program koji izracunava razli¢ite tocke koje ¢e spremati u obliku varijabli x i y. Zatim poziva potprogram
DrawCircleUsingCenter koji je odgovoran za crtanje kruga pomodéu x i y kao sredista.



GraphicsWindow.BackgroundColor = "Black"
GraphicsWindow.PenColor = "LightBlue"
GraphicsWindow.Width = 480
For i = @ To 6.4 Step 0.17

x = Math.Sin(i) * 100 + 200

y = Math.Cos(i) * 100 + 200

DrawCircleUsingCenter()
EndFor

Sub DrawCircleUsingCenter
startX = x - 40
startY =y - 40

GraphicsWindow.DrawEllipse(startX, startY, 120, 120)
EndSub

Slika 46 - Graficki primjer potprograma



Pozivanje potprograma unutar petlji

Ponekad se potprogrami pozivaju unutar petlje, a tijekom tog postupka izvrsavaju isti skup izjava,

ali s razlic¢itim vrijednostima u jednoj ili viSe varijabli. Na primjer, ako imate potprogram pod nazivom
PrimeCheck i taj potprogram odreduje je li broj primarni. MoZete napisati program koji korisniku
omogucuje da unese vrijednost, a zatim pomoéu tog potprograma mozete odrediti je li taj broj
primaran. To ilustrira program u nastavku.

TextWindow.Write("Unesite broj: ")
i = TextWindow.ReadNumber ()
isPrime = "True"
PrimeCheck()
If (isPrime = "True") Then
TextWindow.WriteLine(i +
Else
TextWindow.WriteLine(i +
EndIf

je primarni broj")

nije primarni broj")

Sub PrimeCheck
For j = 2 To Math.SquareRoot (i)
If (Math.Remainder(i, j) = @) Then

isPrime = "False"
Goto EndLoop
EndIf
Endfor
EndLoop:
EndSub

Potprogram PrimeCheck uzima vrijednost i i pokusava je podijeliti manjim brojevima. Ako je i djeljiv

s brojem bez ostatka, to znaci da i nije primarni broj. U tom trenutku potprogram postavlja vrijednost
za isPrime na "False" i zatvara se. Ako broj nije djeljiv pomoéu manjih brojeva, onda vrijednost isPrime
ostaje "True".

B Ch\Users\mjperez\AppData\Local\Temp'tmp55C6.tmp.exe _ C=1Ee X

Unesite broj: 497
49?7 nije primarni bhroj

Press any key to continue...

Slika 47 - Provjera primarnih brojeva



Sada kad imate potprogram koji moZe za vas izvrsiti provjeru primarnih brojeva, pomocu njega mozete,
na primjer, napraviti popis svih primarnih brojeva manjih od 100. Gornji program uistinu je jednostavno
izmijeniti i posti¢i da se naredba PrimeCheck poziva iz petlje. To potprogramu omoguduje da prilikom

svakog izvodenja petlje izraGunava drugaciju vrijednost. Na primjeru u nastavku pogledajmo kako se to radi.

For i = 3 To 100
isPrime = "True"
PrimeCheck()
If (isPrime = "True") Then
TextWindow.WriteLine(i)
EndIf
EndFor

Sub PrimeCheck
For j = 2 To Math.SquareRoot (i)
If (Math.Remainder(i, j) = @) Then

isPrime = "False"
Goto EndLoop
EndIf
Endfor
EndLoop:
EndSub

U gornjem programu vrijednost za i aZurira se prilikom svakog izvodenja petlje. Unutar petlje izvodi se
pozivanje potprograma PrimeCheck. Potprogram PrimeCheck potom uzima vrijednost i te izraCunava je

li i primarni broj. Taj se rezultat sprema u varijabli isPrime kojom se potom pristupa putem petlje izvan
potprograma. Vrijednost za i potom se ispisuje ako je rije¢ o primarnom broju. Buduci da petlja zapocinje
od 3 i nastavlja se sve do 100, dobivamo popis svih primarnih brojeva u rasponu od 3 do 100. U nastavku
je prikazan rezultat programa.

CA\Users\mjperez\AppData‘Local\Temp\tmpEF78.tmp.exe _ C=aia X"

Slika 48 - Primarni brojevi



10. poglavlje

Polja

Sad ste se ve¢ majstor za koristenje varijabli — na koncu konca dospjeli ste ¢ak do ovdje, a jos se uvijek
zabavljate, zar ne?

Idemo se na trenutak vratiti na prvi program koji smo napisali pomo¢u varijabli:

TextWindow.Write("Unesite svoje ime: ")
ime = TextWindow.Read()
TextWindow.WriteLine("Pozdrav" + ime)

U tom smo programu primili i spremili naziv korisnika u varijabli koja se zove ime. Zatim smo kasnije
"pozdravili" korisnika. Pretpostavimo da postoji viSe korisnika, primjerice 5 korisnika. Kako bismo
spremili sva njihova imena? To se moZe napraviti ovako:

TextWindow.Write("Korisnic¢el, unesite svoje ime: ")
imel = TextWindow.Read()
TextWindow.Write("Korisnic¢e2, unesite svoje ime: ")
ime2 = TextWindow.Read()
TextWindow.Write("Korisnice3, unesite svoje ime: ")
ime3 = TextWindow.Read()
TextWindow.Write("Korisnic¢e4, unesite svoje ime: ")
imed4 = TextWindow.Read()
TextWindow.Write("Korisnic¢e5, unesite svoje ime: ")
ime5 = TextWindow.Read()



TextWindow.Write("Pozdrav ")

TextWindow.Write(imel + ",

)
TextWindow.Write(ime2 + ", ")
TextWindow.Write(ime3 + ", ")

)

TextWindow.Write(imed4 + ",
TextWindow.WriteLine(ime5)

Kad pokrenete taj potprogram, dobit cete sljededi rezultat:

- CAUsers\mjperez\AppData\Local\Temp'tmp89F4.tmp.exe

suonje : Shifu
suoje : Tigress
Po

sunje ime:
unesite svoje ime: Oogway
unesite svoje ime: Mantis
Pozdrav Shifu, Tigress, Po. Oogway. Mantis
Press any key to continue...

Slika 49 - Nekoristenje polja

Ocito mora postojati bolji nacin za pisanje tako jednostavnog programa, zar ne? Posebno zato Sto su
racunala uistinu dobra za obavljanje opetovanih zadataka pa zasto bismo se mucili i uvijek iznova pisali
isti kod za svakog novog korisnika? Trik je u tome da se spremi i dohvati vise od jednog imena korisnika
pomocu iste varijable. Ako to moZemo uciniti, onda moZemo koristiti petlju For o kojoj smo govorili

u prethodnim poglavljima. Tu nam priska¢u u pomoc¢ polja.

Sto je polje?

Polje je posebna vrsta varijable koja moze istovremeno sadrzavati viSe vrijednosti. To zapravo znaci da
bismo, umjesto izrade varijabli imel, ime2, ime3, ime4 i ime5 za spremanje pet korisnickih imena, mogli
koristiti samo varijablu ime za spremanje svih pet korisnickih imena. Vise vrijednosti spremamo pomocu
neceg Sto zovemo "indeks". Na primjer, svaka od varijabli ime[1], ime[2], ime[3], ime[4] i ime[5] moZe
spremiti vrijednost. Brojevi 1, 2, 3, 4i 5 zovu se indeksi za polje.

lako ime[1], ime[2], ime[3], ime[4] i ime[5] izgledaju kao razlicite varijable, to je zapravo samo jedna
varijabla. MoZda cete se zapitati koja je prednost toga. Najbolji dio spremanja vrijednosti u polje to je $to
mozZete odrediti indeks pomocu druge varijable, Sto nam omogucuje da jednostavno pristupimo poljima
unutar petlji.



Pogledajmo kako moZzemo upotrijebiti svoje novosteceno znanje za ponovno pisanje prethodnog
programa pomocu polja.

For i =1 To 5
TextWindow.Write("Korisnice" + i +
ime[i] = TextWindow.Read()

EndFor

, unesite svoje ime: ")

TextWindow.Write("Pozdrav ")

For i =1 To 5
TextWindow.Write(ime[i] + ", ")

EndFor

TextWindow.WriteLine("")

Puno je laksSe za Citanje, zar ne? Obratite pozornost na dva podebljana retka. Prvi sprema vrijednost

u polju, a drugi je Cita iz polja. Na vrijednost koju spremite u varijabli ime[1] ne utjeCe ono Sto spremite
u varijabli ime[2]. Stoga ime[1] i ime[2] moZete uglavnom tretirati kao dvije zasebne varijable s istim
identitetom.

© | ChUsers\mjperer\AppDatat\Local\Tempitmp5 742.tmp. S

Korisnicel, unesite svoje ime: Shifu
Korisnice2, unesite svoje ime: Tigress
Korisniced, unesite svoje ime: Po
Korisniced, unesite svoje ime: Qoguway

|Korisnice5, wnesite suoje ime: Mantis
Pozdrav Shifu, Tigress, Po. Qogway. Mantis,.
Press any kev to continue...

Slika 50 - Koristenje polja

Gornji program daje gotovo potpuno isti rezultat kao onaj bez polja, osim zareza na kraju mantise.
To moZete popraviti tako da petlju za ispis napiSete u ovom obliku:

TextWindow.Write("Pozdrav ")

For i =1 To 5
TextWindow.Write(ime[i])
If i < 5 Then

TextWindow.Write(", ")

EndIf

EndFor

TextWindow.WriteLine("")



Indeksiranje polja

U nasem prethodnom programu vidjeli ste da smo koristili brojeve kao indekse za spremanje

i dohvacanje vrijednosti iz polja. Ispada da indeksi nisu ograni¢eni samo na brojeve, a u praksi je
vrlo korisno koristiti i tekstualne indekse. Na primjer, u sljede¢em programu trazimo i spremamo
razli¢ite informacije o korisniku, a zatim ispisujemo informacije koje je korisnik traZio.

TextWindow.Write("Unesite ime: ")
korisnik["ime"] = TextWindow.Read()
TextWindow.Write("Unesite dob: ")
korisnik["dob"] = TextWindow.Read()
TextWindow.Write("Unesite grad: ")
korisnik["grad"] = TextWindow.Read()
TextWindow.Write("Unesite postanski broj: ")
korisnik["postanski broj"] = TextWindow.Read()

TextWindow.Write("Koje informacije zelite? ")
indeks = TextWindow.Read()
TextWindow.WriteLine(indeks + " = " + korisnik[indeks])

# | CA\Users\mjperez\AppData‘\Local\Temp\tmpF73C.tmp.exe

ime: Uijaye
dobh: 38
grad: Redmond
postanski broj:
Koje informacije =zelite? postanski hroj

postanski broj = 98852
Pressz any key to continue...

Slika 51 - KoriStenje nenumerickih indeksa

Vise od jedne dimenzije

Pretpostavimo da Zelite spremiti ime i telefonski broj svih svojih prijatelja, a zatim moci potraZziti
njihove telefonske brojeve uvijek kad je to potrebno — nesto kao telefonski imenik. Kako napisati
takav program?

U tom slucaju rije€ je o dva skupa indeksa

(to se jos naziva dimenzijom polja). Pretpostavimo Indeksi polja ne razlikuju veliko i malo slovo.

da svakog prijatelja prepoznajemo po nadimku. Bas kao kod obicnih varijabli, u pogocima

indeksa polja ne moraju se tocno poklapati

To postaje nas prvi indeks u polju. Nakon Sto
pomocu prvog indeksa dohvatimo varijablu naseg velika i mala slova.
prijatelja, drugi indeksi ime i telefonski broj

omogucit ¢e nam da dohvatimo stvarno ime | telefonski broj tog prijatelja.

Nacin na koji spremamo te podatke izgleda ovako:




prijatelji["Rob"]["Ime"] = "Robert"
prijatelji["Rob"]["Telefon"] = "555-6789"

prijatelji["VI"]["Ime"] = "Vijaye"
prijatelji["V]"]["Telefon"] = "555-4567"

prijatelji["Ash"]["Ime"] = "Ashley"

prijatelji["Ash"]["Telefon"] = "555-2345"
Buduci da imamo dva indeksa u istom polju prijatelji, to se polje naziva dvodimenzionalnim poljem.
Nakon $to postavimo taj program, kao ulaznu vrijednost mozemo uzeti nadimak prijatelja, a zatim

ispisati informacije koje smo spremili o njima. Evo cijelog programa koji to radi:

prijatelji["Rob"]["Ime"] = "Robert"
prijatelji["Rob"]["Telefon"] = "555-6789"

prijatelji["VI"]["Ime"] = "Vijaye"
prijatelji["Vv]"]["Telefon"] = "555-4567"

prijatelji["Ash"]["Ime"] = "Ashley"
prijatelji["Ash"]["Telefon"] = "555-2345"

TextWindow.Write("Unesite nadimak: ")
nadimak = TextWindow.Read()

TextWindow.WriteLine("Ime: " + prijatelji[nadimak]["Ime"])
TextWindow.WriteLine("Telefon: " + prijatelji[nadimak]["Telefon"])

© CAUsersimjperez\AppData\Local\Tempitmp13E.tmp.exe

Unesite nadimak: rob
Ime: Rohert
Telefon: 555-6789

Press any key to continue...

Slika 52 - Jednostavan imenik

Prikaz mreza pomocu polja
Visedimenzionalna polja obi¢no se koriste za prikaz mreza/tablica. MreZe imaju retke i stupce koji se
mogu uklopiti u dvodimenzionalno polje. U nastavku je naveden jednostavan program koji prikazuje

okvire u mreii:



reci = 8
stupci = 8
velic¢ina = 40

For r = 1 To reci
For ¢ = 1 To stupci
GraphicsWindow.BrushColor = GraphicsWindow.GetRandomColor ()
boxes[r][c] = Shapes.AddRectangle(velic¢ina, velicina)
Shapes.Move(boxes[r][c], ¢ * veli¢ina, r * velicina)
EndFor
EndFor

Taj program dodaje pravokutnike i postavlja ih u obliku mreze 8x8. Osim $to slaZe te okvire, on ih
i sprema u polje. To pojednostavnjuje pracenje tih okvira i njihovo ponovno koristenje po potrebi.

i ' SmallBasic Graphics Windo

Slika 53 - Slaganje okvira u mrezu

Primjerice, dodavanjem sljedeéeg koda na kraj prethodnog programa postici ¢e se da se ti okviri
animiraju do gornjeg lijevog kuta.

For r = 1 To reci
For ¢ = 1 To stupci
Shapes.Animate(boxes[r][c], 0, 9, 1000)



Program.Delay(300)
EndFor
EndFor

Slika 54 - Pracenje okvira u mreizi




11. poglavlje
Dogadaji i interaktivnost

U prva dva poglavlja predstavili smo objekte s varijablama svojstva i operacije. Osim svojstava

i operacija, neki objekti imaju varijablu koja se zove dogadaji. Dogadaji su kao signali koji se prikazuju,
na primjer, kao odgovor na korisnic¢ke radnje, na primjer pomicanje misa ili klik miSem. Dogadaji su
donekle suprotnost operacijama. U slu¢aju operacija vi kao programer pozivate ih da bi racunalo
napravilo nesto, dok vam u sluéaju dogadaja racunalo javlja kad se dogodi nesto zanimljivo.

Koliko su korisni dogadaji?

Dogadaiji su od klju¢ne vaznosti za uvodenje interaktivnosti u program. Ako Zelite dopustiti korisniku

da stupi u interakciju s vasim programom, koristit cete dogadaje. Na primjer, ako Zelite napisati program
za igru krizié-kruzi¢. Zelite dopustiti korisniku da odabere Zeli li biti kriZi¢ ili kruZi¢, zar ne? Za to sluze
dogadaji - korisnicke ulazne informacije unutar programa primate pomocu dogadaja. Ako vam to nije
previse jasno, ne brinite. Pogledat ¢emo vrlo jednostavan primjer pomocu kojeg ¢ete shvatiti Sto su
dogadaji i kako se mogu koristiti.

U nastavku je prikazan vrlo jednostavan program koji se sastoji od samo jedne naredbe i jednog
potprograma. Potprogram koristi operaciju ShowMessage da bi na GraphicsWindow objektu
prikazao okvir s porukom za korisnika.

GraphicsWindow.MouseDown = OnMouseDown

Sub OnMouseDown
GraphicsWindow.ShowMessage ("Kliknuli ste.", "Pozdrav")
EndSub



Jedna zanimljivost koju je mogudée primijetiti u vezi s gornjim programom jest redak u kojem se
dodjeljuje naziv potprograma za dogadaj MouseDown objekta GraphicsWindow. Vidjet ¢ete da
MouseDown u velikoj mjeri izgleda kao svojstvo — osim $to mu se umjesto vrijednosti dodjeljuje
potprogram OnMouseDown. To je posebnost dogadaja — kad se dogodi, automatski se poziva
potprogram. U ovom se sluc¢aju potprogram OnMouseDown poziva svaki put kad korisnik misem

klikne na GraphicsWindow. Samo naprijed, pokrenite program i isprobajte ga. Svaki put kad misem
kliknete na GraphicsWindow, prikazat ¢e se okvir s porukom kao $to je onaj prikazan na slici u nastavku.

Pozdrav —E G

Kliknuli ste.

Slika 55 — Odgovor na dogadaj

Takva vrsta upravljanja dogadajem vrlo je napredna i omogucuje prilicno kreativne i zanimljive
programe. Programi napisani na taj nacin ¢esto se nazivaju programima koje pokreéu dogadaiji.

Potprogram OnMouseDown mozete izmijeniti tako da radi druge stvari osim prikaza sko¢nog okvira

s porukom. Na primjer, kao u programu u nastavku, moZete nacrtati velike plave tocke na mjestu gdje
korisnik klikne misem.

GraphicsWindow.BrushColor = "Blue"
GraphicsWindow.MouseDown = OnMouseDown

Sub OnMouseDown
X = GraphicsWindow.MouseX - 10
y = GraphicsWindow.MouseY - 10
GraphicsWindow.FillEllipse(x, y, 20, 20)
EndSub



" SmallBasic Graphics Windo [ =)

Slika 56 - Upravljanje dogadajem pomicanja misa prema dolje

Napominjemo da smo u gornjem programu koristili MouseX i MouseY da bismo dohvatili koordinate
misa. Potom to koristimo da bismo nacrtali krug pomoc¢u koordinata misa kao sredista kruga.

Upravljanje s viSe dogadaja

Zapravo ne postoje ogranicenja u vezi s brojem dogadaja kojim se moZe upravljati. MoZete ¢ak imati
jedan potprogram koji upravlja s viSe dogadaja. No dogadajem mozZete upravljati samo jednom. Ako
istom dogadaju Zelite dodijeliti dva potprograma, prednost ima onaj drugi.

Da bismo to dokazali, uzmimo prethodni primjer i dodajmo potprogram kojim se upravlja pritiskom na
tipke. Isto tako, posti¢i ¢emo da novi potprogram mijenja boju kista pa éete nakon klika misem dobiti
tocku drugacije boje.



GraphicsWindow.BrushColor = "Blue"
GraphicsWindow.MouseDown = OnMouseDown
GraphicsWindow.KeyDown = OnKeyDown

Sub OnKeyDown
GraphicsWindow.BrushColor
EndSub

GraphicsWindow.GetRandomColor ()

Sub OnMouseDown
X = GraphicsWindow.MouseX - 10
y = GraphicsWindow.MouseY - 10
GraphicsWindow.FillEllipse(x, y, 20, 20)
EndSub

# ' SmallBasic Graphics Window

Slika 57 — Upravljanje s viSe dogadaja

Ako pokrenete taj program i kliknete prozor, prikazat ¢e se plava tocka. Ako sada jednom pritisnete
bilo koju tipku i ponovno kliknete, dobit éete tocku drugacdije boje. Nakon pritiska na tipku izvrSava se
potprogram OnKeyDown, a time se boja kista mijenja u sluc¢ajnu boju. Kad nakon toga kliknete misem,
iscrtava se krug pomocu novopostavljene boje, zbog ¢ega nastaju tocke slucajnih boja.



Program za bojanje

NaoruZani dogadajima i potprogramima sad moZemo napisati program koji korisniku omogucuje crtanje

u prozoru. Takav je program iznenadujuée jednostavno napisati, pod uvjetom da problem podijelimo na

manje dijelove. U prvom koraku napisat ¢éemo programu koji omoguduje korisnicima da pomicu misa bilo
gdje u grafickom prozoru te da ostave trag kamo god pomaknu misa.

GraphicsWindow.MouseMove = OnMouseMove

Sub OnMouseMove
X = GraphicsWindow.MouseX
y = GraphicsWindow.MouseY
GraphicsWindow.DrawlLine(prevX, prevyY, x, y)

prevX = X
prevY =y
EndSub

No kad pokrenete taj program, prva linija uvijek pocinje u gornjem lijevom rubu prozora (0, 0). Taj
problem moZemo rijesiti upravljanjem dogadajem MouseDown i biljeZzenjem vrijednosti prevX i prevY
kad se to dogodi.

Isto tako, potrebno nam je da se iscrtavanje izvodi samo ako korisnik drzi pritisnutu tipku misa. Ostatak
vremena liniju nije potrebno iscrtavati. Da bismo to postigli, koristit ¢emo svojstvo IsLeftButtonDown na
objektu Mouse. To svojstvo govori drzi li korisnik pritisnut lijevi gumb. Ako je ta vrijednost istinita, onda
¢emo povuci crtu, a ako nije, crtu ¢emo preskoditi.

GraphicsWindow.MouseMove = OnMouseMove
GraphicsWindow.MouseDown = OnMouseDown

Sub OnMouseDown
prevX = GraphicsWindow.MouseX
prevY = GraphicsWindow.MouseY
EndSub

Sub OnMouseMove
X = GraphicsWindow.MouseX
y = GraphicsWindow.MouseY
If (Mouse.IsLeftButtonDown) Then
GraphicsWindow.DrawLine(prevX, prevY, X, y)

EndIf
prevX = X
prevY =y

EndSub



Dodatak A
Zabavni primjeri

Crtanje fraktala pomoc¢u kornjace

Slika 58 - kornjaca crta fraktal u obliku stabla



kut = 30

delta = 10
udaljenost = 60
Turtle.Speed = 9

GraphicsWindow.BackgroundColor = "Black"
GraphicsWindow.PenColor = "LightGreen"
DrawTree()

Sub DrawTree
If (udaljenost > @) Then
Turtle.Move(udaljenost)
Turtle.Turn(kut)

Stack.PushValue("udaljenost"”, udaljenost)
udaljenost = udaljenost - delta
DrawTree()

Turtle.Turn(-kut * 2)

DrawTree()

Turtle.Turn(kut)

udaljenost = Stack.PopValue("udaljenost™)

Turtle.Move(-udaljenost)
EndIf
EndSub



Fotografije s web-mjesta Flickr

Slika 59 - dohvacanje slika s web-mjesta Flickr

GraphicsWindow.BackgroundColor = "Black"
GraphicsWindow.MouseDown = OnMouseDown

Sub OnMouseDown
slika = Flickr.GetRandomPicture("planine, rijeke")

GraphicsWindow.DrawResizedImage(slika, @, 0, 640, 480)
EndSub

Dinamicna pozadina radne povrsine

Za i =1 do 10
slika = Flickr.GetRandomPicture("planine")
Desktop.SetWallPaper(slika)
Program.Delay(10000)

EndFor




Igra s lopticom

GraphicsWindow.BackgroundColor
palica = Shapes.AddRectangle(120, 12)
loptica = Shapes.AddEllipse(16, 16)

GraphicsWindow.MouseMove

# " SmallBasic Graphics Window

X = 0

y =29
deltaX =1
deltay =1
RunLoop:

X = X + deltaX
y =y + deltaYy

gw = GraphicsWindow.Width
gh = GraphicsWindow.Height
If (x >= gw - 16 or x <= @) Then

deltaX = -deltaX
EndIf
If (y <= @) Then
deltaY = -deltaY
EndIf

Slika 60 - igra s palicom

"DarkBlue"

= OnMouseMove




padX = Shapes.GetLeft (palica)

If (y = gh - 28 and x >= padX and x <= padX + 120) Then
deltaY = -deltaY

EndIf

Shapes.Move(loptica, x, y)
Program.Delay(5)

If (y < gh) Then
Goto RunLoop
EndIf

GraphicsWindow.ShowMessage("Izgubili ste", "Palica")

Sub OnMouseMove

paddleX = GraphicsWindow.MouseX

Shapes.Move(palica, paddleX - 60, GraphicsWindow.Height - 12)
EndSub



Dodatak B

Boje

Slijedi popis boja koje podrzava Small Basic, kategoriziranih prema osnovnoj boji.

Nijanse crvene boje . . .
Nijanse narancaste boje

IndianRed #CD5C5C

Coral H#FF7F50

Tomato #FF6347

OrangeRed #FF4500

e DarkOrange #FF8CO0

Crimson #DC143C

Orange #FFA500
Red #FFO000
FireBrick #B22222 Nijanse Zute boje
DarkRed #8B0000 Gold #FFD700

Yellow #FFFFOO

Nijanse ruzicaste boje :

LightYellow #FFFFEO
Pink #FFCOCB

LemonChiffon #FFFACD
LightPink #FFB6C1

LightGoldenrodYellow | #FAFAD2
HotPink #FF69B4

PapayaWhip H#FFEFD5
DeepPink #FF1493

Moccasin #FFE4B5
MediumVioletRed #C71585

PeachPuff #FFDAB9

PaleVioletRed #DB7093




PaleGoldenrod HEEESAA

Khaki H#FOEG8C
DarkKhaki #BDB76B MediumSeaGreen #3CB371

SeaGreen #2E8B57

Nijanse ljubicaste boje

ForestGreen #228B22
Lavender H#EGEGFA

Green #008000

DarkGreen #006400

YellowG H#9ACD32
Violet HEES2EE I (R

: OliveDrab #6B8E23
Orchid #DA70D6

. Olive #808000
Fuchsia H#FFOOFF e — A —

DarkOliveGreen #556B2F
Magenta H#FFOOFF

MediumOrchid #BA55D3

. DarkSeaGreen #8FBC8F
MediumPurple #9370DB

. LightSeaGreen #20B2AA
BlueViolet H#8A2BE2 | o |

. DarkCyan #008B8B
DarkViolet #9400D3 ——

Teal #008080

DarkOrchid #9932CC

DarkMagenta #8B00SB Nijanse plave boje
Purple #800080

Indigo #4B0082
SlateBlue #6A5ACD LightCyan HEOFFFF

DarkSlateBlue #483D8B PaleTurquoise H#AFEEEE

MediumSlateBlue #7B68EE Aguamarine #7FFFD4

Nijanse zelene boje

MediumTurquoise #48D1CC
GreenYellow #ADFF2F

DarkT i #00CED1
Chartreuse HTFFFOO Sl Bl

B E

LawnGreen #7CFC00 CRGEEBILE #SFSEAC

SteelBlue #4682B4

LightSteelBlue #BOC4DE

P Bl #BOEOE
PaleGreen #98FB9S owderBlue OEOE6

LightBlue #ADDSE6
LightGreen #90EE90 -

SkyBlue #87CEEB




LightSkyBlue #87CEFA Nijanse bijele boje

DeepSkyBlue #OOBFFF White HFFEFFE
DodgerBlue #1E9OFF Snow #FFFAFA
CornflowerBlue #6495ED Honeydew #HFOFFFO
MediumSlateBlue #7B6SEE MintCream H#FSFFFA
RoyalBlue #4169E1 Azure #FOFFFF
Blue #0000FF AliceBlue #FOFSFF
MediumBlue #0000CD GhostWhite HFSFSFF
DarkBlue | #00008B | WhiteSmoke HFSFSES
FW Seashell HFFFSEE
MidnightBlue #191970 Beige #F5F5DC
L OldLace HFEDF5E6
Nijanse smede boje
FloralWhite HFFFAFO
Cornsilk H#FFFSDC
Ivory #FFFFFO
BlanchedAlmond H#FFEBCD
AntiqueWhite H#FAEBD7
Bisque HFFE4C4
Linen H#FAFOE6
NavajoWhite #FFDEAD
LavenderBlush H#FFFOF5
Wheat #F5DEB3
MistyRose #FFEAE1

Nijanse sive boje

RosyBrown HBCSFSF Gainsboro #DCDCDC

LightGray #D3D3D3

Goldenrod #DAA520 Silver #COCOCO

DarkGoldenrod #B8860B

Peru #CD853F Gray #808080

DimGray #696969

Chocolate #D2691E

SaddleBrown #8B4513

LightSlateGray #778899

Sienna #A0522D SlateGray #708090

Brown #AS52A2A DarkSlateGray HOFAFAF

Black #000000

Maroon #800000



